# ENCM 369 Winter 2023 Lab 3 for the Week of January 30

Steve Norman
Department of Electrical & Software Engineering
University of Calgary

January 2023

## Administrative details

### Each student must hand in their own assignment

Later in the course, you may be allowed to work in pairs on some assignments.

### Due Dates

The Due Date for this assignment is 6:00pm Friday, February 3.
The Late Due Date is 6:00pm Saturday, February 4.

The penalty for handing in an assignment after the Due Date but before the Late Due Date is 3 marks. In other words, X/Y becomes (X–3)/Y if the assignment is late. There will be no credit for assignments turned in after the Late Due Date; they will not be marked.

### Marking scheme

| | |
|---|---|
| A | 1 mark |
| C | 8 marks |
| E | 8 marks |
| F | 3 marks |
| total | 20 marks |

### How to package and hand in your assignments

You must submit your work as a *single PDF file* to the D2L dropbox that will be set up for this assignment. The dropbox will be configured to accept only file per student, but you may replace that file as many times as you like before the due date.

See the Lab 1 instructions for more information about preparing and uploading your PDF file.

## Getting `.c` and `.asm` files for this lab

The files you need for all six exercises are in a folder called `encm369w23lab03`, which has been uploaded to D2L in `.zip` format. Before starting work, you must download the `.zip` file, and extract its contents in a suitable place in the file system of your computer.

# Exercise A: Instructions that try to do bad things

## Read This First

The point of this exercise is to help you understand what is going on when RARS programs fail to run to completion.

Here is a little background. Some instructions may ask for behaviour the processor can not deliver, depending on what is in the registers used by the instruction. Here are two examples:

- `lw s1, 0(s0)` is attempted, but the address in `s0` is not a multiple of four. With MIPS hardware, words cannot be read from (or written to) addresses that aren't multiples of four.
- `lw s1, 0(s0)` is attempted, but the address in `s0` is zero. Zero is a multiple of four, but programs don't have access to memory at address zero. (This is like trying to access data through a null pointer in C.)

In a real RISC-V processor, what happens in the above cases—and many other out-of-the-ordinary situations—is that an *exception* occurs. An exception causes the program to be suspended while *exception handling* code is run. (This is a very vague description; exceptions will be covered in much more detail later in the course.)

In RARS, by default, this kind of exception causes immediate termination of a program, with the offending instruction highlighted in the text segment. That's quite useful for debugging—it lets you examine the exact state of registers and memory at the moment things went wrong.

## What to Do

In RARS, try assembling and running all of the three programs in the folder `encm369w23lab03/exA`. Before each run, use the Clear button to clear old messages from the Messages tab.

Write down the messages that appear in the Messages tab for each of the three program runs.

## What to Include in Your PDF Submission

Include a nicely organized list of all of the messages you were asked to write down.

# Important notes about RARS programming

*Please read this section carefully, because it has information you need to know to read and write assembly language code in Exercises B–F.*

## Calling conventions

When you think about calling conventions, *always imagine a program with hundreds of functions containing thousands of function calls!* When coding any one function, it is crucial to have a set of rules that allows you to make that function compatible with the rest of the program, without having to look at the code for any other function.

*Even though programs you write in this course will only have a few relatively small functions, you are required to follow the calling conventions.*

Here are the rules seen so far in the course:

- Arguments are passed in `a0`, `a1`, ..., `a7`, in that order.
- The return value, if there is one, goes in `a0`.
- On exit from a function, `sp` must contain the same address it contained on entry to that function.
- Nonleaf functions should save `ra` on the stack in the prologue and restore it in the epilogue. There is no need to do this in a leaf function.
- Any function, leaf or nonleaf, should save any s-registers it uses on the stack in the prologue and restore them in the epilogue.
- Nonleaf functions must *not* rely on t-registers to maintain their values across function calls.
- Nonleaf functions must separate incoming arguments from outgoing arguments. There are two ways to do this:
  - In the prologue, copy incoming arguments to stack slots.
  - In the prologue, copy incoming arguments to s-registers.

  There is *no need* to restore incoming argument values to a-registers in the epilogue.

## How large should a stack frame be?

This question is more complicated than it might seem. The answer will depend on the context in which the question is asked!

**Examples in Section 6.3.7 of the textbook and in ENCM 369 lecture and tutorial examples:** *The size of a stack frame should be just big enough to hold all the words that will be packed into it.* This is simple, and also allows for compact, easy-to-draw diagrams.

**RARS programming in ENCM 369 labs:** *The size of a stack frame should be a multiple of 8 words—so, a multiple of 32 bytes.* So, when deciding how big a stack frame should be, use the following rules:

| number of words needed | frame size in bytes |
|---|---|
| 1–8 | 32 |
| 9–16 | 64 |
| 17–24, 25–32, etc. | 96, 128, etc. |

Here is why: If you follow the above rule, and if the address in `sp` is a multiple of 32 when `main` starts, then every stack frame will occupy exactly one or two (or more) rows in the RARS Data Segment window. This "wastes" some stack space but makes it relatively easy to locate words within stack frames, as illustrated in Figure 1.

## How to allocate local variables

It was suggested early in the course that s-registers are to be allocated for local variables of `int` or pointer type. *In fact, this suggestion is too simple*—in some cases it just won't work, and in others it leads to minor inefficiencies. Here is a better set of rules:

- In a *nonleaf* function, the best choice for a local variable of `int` or pointer type is usually an s-register.

**Figure 1:** Two ways to visualize an 8-word stack frame. Above: How it would be drawn in a lecture or textbook example. Below: How it would appear in the RARS Data Segment window, assuming that *all* stack frame sizes are multiples of 8 words.




| Address | Value (+ 0) | Value (+ 4) | Value (+ 8) | Value (+ c) | Value (+ 10) | Value (+ 14) | Value (+ 18) | Value (+ 1c) |
|---|---|---|---|---|---|---|---|---|
| ⋮ | ⋮ | ⋮ | ⋮ | ⋮ | ⋮ | ⋮ | ⋮ | ⋮ |
| | | | | | | | | |
| ⋮ | ⋮ | ⋮ | ⋮ | ⋮ | ⋮ | ⋮ | ⋮ | ⋮ |

8-word stack frame. Offsets (in hex) are in the header row.

Base address of stack frame can be found here.

- In a *leaf* function, the best choice for a local variable of `int` or pointer type is usually a t-register. By choosing a t-register, you avoid the need to save and restore its value to and from the stack.
- But some local variables can't be allocated in registers at all. An array must be in memory. An individual `int` variable whose address is taken with the C `&` operator must be in memory. If a local variable of a function must be in memory, space should be allocated within that function's stack frame for that variable. (This item is relevant to Lab 4 more than Lab 3; in Lab 3 you won't have to put any local variables in memory.)

## Global symbols and the `.globl` directive

The assembler directive

```
.globl  foo
```

says that `foo` is to be treated as a *global symbol*.

In a real assembly language development environment global symbols are needed when programs are built from multiple source files. (*Source files* in software development are the text files edited by programmers as they work on a program. So in C development, the source files are the `.c` and `.h` files produced by programmers, but in assembly language development, source files are text files containing assembly language code—for RARS, `.asm` files.) Code and data labeled with global symbols can be accessed from all source files belonging to the program. This is necessary when a function in one source file calls a function in another source file, and when a function in one source file accesses an external variable set up in another source file.

RARS allows you to build programs from multiple source files, and in that case, global symbols are necessary. However, for the small programs you will work with in ENCM 369, it is easier to put all the code for a program in a single `.asm` file.

But there is another good reason to make some of your RARS symbols global. In the RARS Labels window, global symbols are listed separately from non-global symbols. It is helpful to make your labels for functions and external variables global, because the addresses for those labels are probably more interesting than addresses of various labelled instructions in loops and if-statements.

# Exercise B: Looking at memory accesses in RARS

## Read This First

There are no marks for this exercise, but doing it will show you some things about RARS that will help with many later lab exercises.

## What to Do

Bring the file `encm369w23lab03/exB/ex3B.asm` into the RARS Edit tab, and assemble it.

In the Execute tab, put a check in the Bkpt (breakpoint) column beside the instruction at address `0x00400088`, in the Text Segment window. This is the `sw` instruction in `main` that copies the incoming `ra` value to the stack frame of `main`. Click the Run icon to start the program; you should find the program paused at the breakpoint you just set.

Click the Single-Step icon so that the processor will read and execute the `sw` instruction.

You should see that the control you've previously seen as


has changed to




That change was automatic as a result of the `sw`, but sometimes you may need to switch manually back and forth between viewing memory starting at `0x10010000` and at current sp.

You should also see that `0x0040000c`, the return address from `main` back to the start-up code, has been copied from `ra` into the stack at an offset of 20 bytes (0x14 bytes) from where `sp` points.

Slowly single-step through the rest of program execution. Here are some things to watch for:

- The instruction `jal fill` writes `0x00400054` into the PC to get `fill` started, and writes `0x00400090` into `ra` to provide a path back from `fill` to exactly the right place in `main`.
- `fill` writes a few words into the `.data` area. Note the automatic change from

  
  

  back to


If you want to see the stack again, you will have to operate the above control manually.

- The `lw` instruction near the end of `main` copies `0x0040000c` back into `ra` to allow a return back to the start-up code.

### What to Include in Your PDF Submission

Nothing.

## Exercise C: Translating a simple program with procedure calls

### What to Do

Study the program in the file `encm369w23lab03/exC/functions.c`.

Follow the instructions found in the comments in the C code. To get started, make a copy of the file `stub1.asm`, which is in the same folder as `functions.c`. Use `functions.asm` as the name for your copy.

Your assembly language procedures must follow all calling conventions introduced so far in ENCM 369.

Note that `main` is to be treated *just like any other procedure*. That means if it uses s-registers, it must save the old s-register values in the prologue and restore them in the epilogue. It does *not* matter that you can tell from reading the start-up and clean-up code that this save-and-restore is unnecessary—the point is to get as much practice as possible following the usual conventions for coding procedures.

### What to Include in Your PDF Submission

*Include a listing of your completed RARS program.* Make sure your listing will be *easy* for your teaching assistants to read.

## Exercise D: Detailed study of an assembly-language program

### Read This First

There are no marks for this exercise, because as you will see, it is easy to check all the answers by running the program in RARS. *However, it is a very important exercise. You can expect to find a similar (but shorter and simpler) problem on Midterm #1.*

This exercise is designed to help solidify your understanding of exactly how registers and stack memory are used to allow assembly-language functions to work together.

### What to Do, Part I

*Do not use RARS at all in this part of the exercise!*

Print the figures on pages 9 and 10 on separate pieces of paper (not back-to-back on a single sheet).

The assembly-language code in Figure 3 is a correct (carefully tested) translation of the C code in the same Figure.

The program will pass through POINT ONE three times. You are asked to determine the state of the program the *third time* it gets to POINT ONE.

Specifically, your goal is to fill in *all* of the empty boxes for registers and memory words in Figure 4 with either *numbers* or the word "unused" to indicate that the program has neither read nor written a particular memory word.

For any given register or memory word, use either base ten or hexadecimal notation, whichever is more convenient. (For example, the numbers 200 and 1000 will appear in memory words somewhere; there is no need to convert either of them to hex.)

To get you started, the stack frame of `main` has been filled in for you. Note the saved `ra` and `s0` values, and the six unused words in the 8-word stack frame.

Here are some important hints:

- It is possible to solve the problem by ignoring the C code, and tracing the effect of every assembly-language instruction starting from the first instruction of `main`. However, that method is slow and painful.

  *It is much better to use clues given in the C code.*

  Example: You can see that `main` calls `procA`, `procA` calls `procB`, and `procB` calls `procC` from within a loop. Also, `procC` is leaf, so probably won't have a stack frame at all. With that information gained from the C code, you can then look at the *prologues* of the assembly language functions to determine sizes and layouts of stack frames, and mark out the frames of `procA` and `procB` on the diagram of the stack.

  Another example: You can use the C code to determine which words in the array `gg` are pointed to by `p` and `q`. You can see from the assembly language prologue for `procB` that `s0` and `s1` are used for `p` and `q`. Because `procC` does not use `s0` or `s1`, finding the addresses in `p` and `q` will give you the contents of `s0` and `s1` at POINT ONE.

- See Figure 2 for a good method of finding function return addresses.

## What to Do, Part II

Use the file `encm369w23lab03/exD/ex3D.asm` to check your Part I solution using RARS. Set a breakpoint on the first `sll` instruction in `procC`, and run the program until that breakpoint has been hit for the third time. Use




to look at the stack, and


to look at the array `gg`. It might (or might not, depending on exactly how you use RARS) be helpful to use

to get all three stack frames visible at the same time. Uncheck


if you would like to see numbers in base ten; check it again for hexadecimal display.

**Figure 2:** A quick way to determine return addresses. In this example the return address from `procA` back to `main` is calculated—it's `0x0040_013c`. Notes: skip lines without instructions; count two real instructions for the pseudoinstruction `la` *GPR*, *label*.

```
main:
    addi    sp, sp, -32       # 0x0040_0118
    sw      ra, 4(sp)         #       _011c
    sw      s0, 0(sp)         #       _0120

    addi    s0, zero, 1000    #       _0124
    addi    a0, zero, 200     #       _0128
    la      a1, gg            #  _012c, _0130
    addi    a2, zero, 3       #       _0134
    jal     procA             #       _0138
    add     s0, s0, v0        #       _013c
    add     v0, zero, zero

    lw      s0, 0(sp)
    lw      ra, 4(sp)
    addi    sp, sp, 32
    jr      ra
```

## What to Include in Your PDF Submission

Nothing.

**Figure 3:** C and assembly language code listings for Exercise D.

```
int procC(int x)
{
    // POINT ONE

    return 8 * x + 2 * x;
}

void procB(int *p, int *q)
{
    while (p != q) {
        *p = procC(*p);
        p++;
    }
}

int procA(int s, int *a, int n)
{
    int k;
    k = n - 1;
    procB(a, a + n);
    while (k >= 0) {
        s += a[k];
        k--;
    }
    return s;
}

int gg[] = { 2, 3, 4 };

int main(void)
{
    int mv;
    mv = 1000;
    mv += procA(200, gg, 3);
    return 0;
}
```

```
    .text
    .globl  procC
procC:
    # POINT ONE

    slli    t0, a0, 3
    slli    t1, a0, 1
    add     a0, t0, t1
    jr      ra

    .text
    .globl  procB
procB:
    addi    sp, sp, -32
    sw      ra, 8(sp)
    sw      s1, 4(sp)
    sw      s0, 0(sp)
    add     s0, a0, zero
    add     s1, a1, zero

L1:
    beq     s0, s1, L2
    lw      a0, (s0)
    jal     procC
    sw      a0, (s0)
    addi    s0, s0, 4
    j       L1
L2:
    lw      s0, 0(sp)
    lw      s1, 4(sp)
    lw      ra, 8(sp)
    addi    sp, sp, 32
    jr      ra
```

```
    .text
    .globl  procA
procA:
    addi    sp, sp, -32
    sw      ra, 16(sp)
    sw      s3, 12(sp)
    sw      s2, 8(sp)
    sw      s1, 4(sp)
    sw      s0, 0(sp)
    add     s0, a0, zero
    add     s1, a1, zero
    add     s2, a2, zero

    addi    s3, s2, -1
    add     a0, s1, zero
    slli    t0, s2, 2
    add     a1, s1, t0
    jal     procB
L3: blt     s3, zero, L4
    slli    t1, s3, 2
    add     t2, s1, t1
    lw      t3, (t2)
    add     s0, s0, t3
    addi    s3, s3, -1
    j       L3
L4: add     a0, s0, zero

    lw      s0, 0(sp)
    lw      s1, 4(sp)
    lw      s2, 8(sp)
    lw      s3, 12(sp)
    lw      ra, 16(sp)
    addi    sp, sp, 32
    jr      ra

    .data
    .globl  gg
gg: .word   2, 3, 4

    .text
    .globl  main
main:
    addi    sp, sp, -32
    sw      ra, 4(sp)
    sw      s0, 0(sp)

    addi    s0, zero, 1000
    addi    a0, zero, 200
    la      a1, gg
    addi    a2, zero, 3
    jal     procA
    add     s0, s0, a0
    add     a0, zero, zero

    lw      s0, 0(sp)
    lw      ra, 4(sp)
    addi    sp, sp, 32
    jr      ra
```

**Figure 4:** Worksheet for Exercise D.

Addresses of functions and data:

| label | address |
|---|---|
| procC | 0x0040_0050 |
| procB | 0x0040_0060 |
| procA | 0x0040_00a4 |
| main | 0x0040_0118 |
| gg | 0x1001_002c |

Contents of some GPRs when `main` starts:

| GPR | contents |
|---|---|
| s0-s11 | all 0 |
| sp | 0x7fff_efe0 |
| ra | 0x0040_000c |



# Exercise E: More practice with functions

## What to Do

Study the C source file in the folder `encm369w23lab03/exE`

Follow the instructions found in the comments in the C code. You can start by making a copy of `stub1.asm` from Exercise C.

Your assembly language functions must follow all calling conventions introduced so far in ENCM 369.

## What to Include in Your PDF Submission

*Include a listing of your completed RARS program.*

# Exercise F: A function to swap contents of integer variables

## What to Do

Study the `.c` and `.asm` files in `encm369w23lab03/exF`

Follow the instructions found in the comments in the C code. Your assembly language functions must follow all calling conventions introduced so far in ENCM 369.

## What to Include in Your PDF Submission

*Include a listing of your completed RARS program.*